Problema Entrega 1

Un sólido deformable tiene forma de pirámide regular con base cuadrada de lado a y altura $h = \frac{a}{\sqrt{5}}$. Su centro está colocado en el origen de un sistema cartesiano de coordenadas y los lados de la base son paralelos a los ejes x, y. El tensor de tensiones en la pirámide tiene una expresión matricial en el sistema xyz que es

$[T] = \{\{k\, x\text{^}2, 0, k\, z\text{^}2\}, \{0, 0, 0\}, \{k\, z\text{^}2, 0, -6\, k\, a\text{^}2\}\}$

Se pide
-- Calcular el vector tensión en el punto B, baricentro del triángulo con vértices (l/2,l/2,0), (-l/2,l/2,0) y (0,0,h), indicando cuáles son sus componentes intrínsecas.

-- Dibujar el diagrama de Mohr del tensor de tensiones en el punto B, indicando cuál es la tención tangencial máxima en dicho punto si h = 1m y k = 9 N/m^4

-- Determinar el factor de seguridad según los criterios de Tresca y von Mises del estado tensional en B si σe = 400 MPa.

-- Calcular la fuerza total que se realiza sobre la base de la pirámide

Estado tensional

El estado tensinal está definido por el campo del tensor de tensiones para todo punto (x, y, z)

```
In[1]:= T[x_, y_, z_] := {{k x^2, 0, k z^2}, {0, 0, 0}, {k z^2, 0, - 6 k a^2}};
        T[x, y, z] // TraditionalForm
```

Out[2]//TraditionalForm=

$$\begin{pmatrix} k\,x^2 & 0 & k\,z^2 \\ 0 & 0 & 0 \\ k\,z^2 & 0 & -6\,a^2\,k \end{pmatrix}$$

Punto B y normal a la pirámide en dicho punto

El punto B está en el baricentro de la cara con vertices que tiene coordenadas r1, r2, r3

```
In[3]:= r1 = {a / 2, a / 2, 0};
        r2 = {-a / 2, a / 2, 0};
        r3 = {0, 0, h};
        B = 1 / 3 (r1 + r2 + r3) /. a → Sqrt[5] h
```

Out[6]= $\left\{0, \frac{\sqrt{5}\,h}{3}, \frac{h}{3}\right\}$

Para calcular el vector normal unitario saliente a la cara de la pirámide encontramos primero un vector saliente perpendicular a ésta, que llamamos v y después lo normalizamos

```
In[7]:= v = Cross[r2 - r1, r3 - r1]
```

Out[7]= $\left\{0, a\,h, \frac{a^2}{2}\right\}$

In[8]:= n = Simplify$\left[\frac{v}{\text{Norm}[v]}\right.$ /. a → Sqrt[5] h, Assumptions → {h > 0, a > 0}$\left.\right]$

Out[8]= $\left\{0, \frac{2}{3}, \frac{\sqrt{5}}{3}\right\}$

Vector tensión en B, componentes intrínsecas

El vector tensión en B se obtiene simplemente evaluando T y después aplicando este tensor sobre el vector unitario calculado anteriormente.

In[9]:=
```
t = Simplify[T[B[[1]], B[[2]], B[[3]]].n  /. a → Sqrt[5] h, Assumptions → h > 0]
```

Out[9]= $\left\{\frac{1}{27}\sqrt{5}\, h^2 k,\ 0,\ -10\sqrt{5}\, h^2 k\right\}$

In[10]:=
```
σn = t.n
```

Out[10]= $-\frac{50\, h^2 k}{3}$

In[11]:=
```
τ = Simplify[Sqrt[t.t - σn^2] , Assumptions → h > 0]
```

Out[11]= $\frac{1}{27}\sqrt{162005}\, h^2 \sqrt{k^2}$

Estado tensional en B

In[12]:=
```
TT = Simplify[T[B[[1]], B[[2]], B[[3]]]  /. {a → Sqrt[5], h → 1, k → 9}] ;
```

In[13]:=
```
TT // TraditionalForm
```

Out[13]//TraditionalForm=

$$\begin{pmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 1 & 0 & -270 \end{pmatrix}$$

In[14]:=
```
ev = N[Eigenvalues[TT]]
```

Out[14]= {-270.004, 0.00370365, 0.}

El valor máximo de la tensión de cortante no lo hallamos gráficamente, sino directamente a partir de los valores de las tensiones principales.

In[15]:=
```
τmax = Simplify[(Max[ev] - Min[ev]) / 2]
```

Out[15]= 135.004

Factores de seguridad

Los factores de seguridad según los criterios de Tresca y von Mises se obtienen a partir de las tensiones equivalentes de ambos casos.

In[16]:=
```
σe = 400;
σeq = Max[ev] - Min[ev];
Ntresca = N[σe / σeq]
```

Out[18]= 1.48144

In[19]:=
```
σmises[ev_] := Sqrt[1 / 2 ((ev[[1]] - ev[[2]]) ^2 + (ev[[2]] - ev[[3]]) ^2 + (ev[[3]] - ev[[1]]) ^2)];
```

In[20]:=
```
σeq = N[σmises[ev]]
```

Out[20]= 270.006

In[21]:=
```
Nmises = N[σe / σeq]
```

Out[21]= 1.48145

Fuerza sobre la base

La fuerza sobre la base se obtiene de integrar el vector tensión sobre todos los puntos de ésta. Como el vector tensión es constante, la integral es inmediata.

```
In[22]:= t = Simplify[T[x, y, 0].{0, 0, -1}]
```

Out[22]= $\{0, 0, 6\, a^2\, k\}$

```
In[23]:= F = t * a^2
```

Out[23]= $\{0, 0, 6\, a^4\, k\}$